ESUG meeting 2012

An application of funnel plots to monitoring loan defaults in consumer credit portfolios

Yevhen (Eugene) Yankovskyy,
ATB Financial, Edmonton, Canada
yyankovskyy@atb.com

Edmonton
2012

Disclaimer

The views and opinions expressed here are those of the author and do not reflect the official policy or position of the employer.

Presentation plan

- Research problem
- Solution description
- Application example
- Conclusions

Research problem

30-, 60-, 90-day delinquency rates are conventional indicators for monitoring default situation in credit portfolios.

- Need in statistically reasonable and effective tool

a) to compare the default rates in areas with different portfolio volumes (sample sizes);

b) to make a qualitative inference about the default rate deviations from the overall rate average, given their portfolio volumes (sample sizes);

c) to monitor default situation in the neighbouring areas.

Solution: algorithm

1. Develop several funnel plots at meaningful confidence levels
2. Classify the areas by deviation of their default rates from the overall average at the funnel plots
3. Transfer the qualitative results onto maps

Monitoring defaults: Funnel plots

$$\hat{p} = \frac{1}{K}\sum_{i=1}^{K}\frac{D_i}{n_i}$$

$$I_{[0;1]}\left(\hat{p} \pm Z_{(1-\alpha/2)}\sqrt{\frac{\hat{p}(1-\hat{p})}{n}}\right)$$

$$n \in [\min(n_i), \max(n_i)]$$

where

D_i = number of defaulted loans in i area of interest, i = 1,...,K;

$\hat{p}$ is a population default rate estimate;

n_i = a portfolio (sample) size in area i.

Default monitoring system: Classification system

Risk level	Statistic confidence interval boundaries	Colour (SAS colour)	Suggested credit policy applied in the area
Extremely high	95% and higher	Red	Emergency intervention: stop lending in the area, investigation of the situation
Moderate High	90-95%	Pink	Toughening the credit policies: increasing credit limits and down payment sizes; higher customers' risk score requirements
Normal	10-90%	Green (lime)	No change in lending policies, keep monitoring
Moderate Low	5%-10%	Light blue (cyan)	Loosening the credit policies: slightly lower credit limits, down payment sizes; lower customers' risk score requirements
Extremely Low	5% and lower	Blue	Applying aggressive risk taking credit policies: significantly lower credit limits, down payment sizes, lower customers' risk score requirements

Application example: setting

Portfolio data were randomly generated for the United States:

90-day delinquency rate$_i$ ~ Beta(α=0.98,β=97.02)

Loan volume$_i$ ~ 10* Neg-Binomial(p=0.05, k=2)

i = 1,..,52

Application example: funnel plots



Application example: Classification results

Table. The US states with high delinquency rates

State name abbreviation	Name of State	Loan volume	Observed delinquency rate	90% delinquency rate bound
TN	Tennessee	800	.01612	.01508
MT	Montana	640	.01719	.01562
ND	North Dakota	540	.01615	.01608
CA	California	440	.02467	.01668
MS	Mississippi	400	.03951	.01698
LA	Louisiana	330	.01997	.01764
MO	Missouri	310	.02825	.01787
HI	Hawaii	200	.02377	.01968
MN	Minnesota	170	.02720	.02046
CT	Connecticut	80	.02817	.02504

Application example: Classification results

Table. The US states with low delinquency rates

State name abbreviation	Name of State	Loan volume	Observed delinquency rate	10% delinquency bound
WA	Washington	1490	.00444	.00709
OR	Oregon	800	.00526	.00586
KS	Kansas	710	.00493	.00557
SC	South Carolina	710	.00313	.00557
AL	Alabama	560	.00231	.00496
CO	Colorado	530	.00022	.00480
NE	Nebraska	460	.00026	.00439
VT	Vermont	460	.00319	.00439
AZ	Arizona	450	.00221	.00432
KY	Kentucky	450	.00313	.00432
IA	Iowa	440	.00194	.00425
DE	Delaware	420	.00310	.00411
MD	Maryland	410	.00132	.00403
OK	Oklahoma	240	.00112	.00205

Application example: map

Chart. Classification of credit card portfolios by 90−day delinquency rate in the USA



Key SAS code: Uploading SAS maps

```
libname maps 'C:\Program Files\SAS\SASFoundation\9.2\maps\';

/* Generating random values for volume and delinquency rates at the state level */
data usmap;
   set maps.us2 (keep = STATE STATECODE STATENAME);
   do i =1 to 52;
              dlqrt=rand('BETA',0.98,97.02);
    volume=10*rand('NEGBINOMIAL',0.05,2);
   end;
 drop i;
 label dlqrt='Observed delinquency rate';
 label volume='Loan volume';
 label STATECODE='State name abbreviation';
run;

proc sort data=usmap; by STATECODE; run;
```

Key SAS code: calculating funnel plots

```
proc sql;
    create table  temp.usmap as
    select *,
    avg(dlqrt) as dlqrt_us label='US average delinquency rate',
    /*90% CI: 5%-95% boundaries:  */
    calculated dlqrt_us+1.645*sqrt(calculated dlqrt_us*(1-calculated dlqrt_us)/volume) as dlqrt_95pct
    label='95% delinquency rate boundary',
    case when (calculated dlqrt_us-1.645*sqrt(calculated dlqrt_us*(1-calculated dlqrt_us)/volume))<0 then 0
         when (calculated dlqrt_us-1.645*sqrt(calculated dlqrt_us*(1-calculated dlqrt_us)/volume))>1 then 1
          else calculated dlqrt_us-1.645*sqrt(calculated dlqrt_us*(1-calculated dlqrt_us)/volume)
    end as dlqrt_5pct label='5% delinquency rate bound',
    /*Similar approach for 80% CI: 10%-90% boundaries: 1.281 */
    /* Classifying in risk levels */
    case when (calculated dlqrt_95pct)<=dlqrt then 'Extremely high'
     when (calculated dlqrt_90pct)<=dlqrt<(calculated dlqrt_95pct) then 'Moderate high'
              when (calculated dlqrt_10pct)<dlqrt<(calculated dlqrt_90pct) then 'Normal'
              when (calculated dlqrt_5pct)<dlqrt<=(calculated dlqrt_10pct) then 'Moderate low'
             else 'Extremely low'
             end as risk_level
from  usmap;
quit;
```

Key SAS code: Indetifying outliers

```
data temp.outliers;
     set temp.usmap (keep= STATECODE STATENAME volume dlqrt dlqrt_90pct dlqrt_10pct);
     where dlqrt>=dlqrt_90pct or dlqrt<=dlqrt_10pct;
     format dlqrt 6.5 dlqrt_90pct 6.5 dlqrt_10pct 6.5 flag_print $CHAR15.;
     select;
        when(dlqrt>=dlqrt_90pct) flag_print='Up-outlier';
        when(dlqrt<=dlqrt_10pct) flag_print='Down-outlier';
                otherwise flag_print='Non-outlier';
       end;
   XSYS = '2';YSYS = '2';
   X = volume;Y = dlqrt;
   FUNCTION='LABEL';
 function='label';
   text=STATECODE;
   size=1;
   position='1';color='red';
  output;
run;
```

Key SAS code: constructing a funnel plot

```
proc sort data=temp.outliers;
by flag_print descending volume; run;

/* Developing a funnel plot */
proc sort data=temp.usmap;
by volume; run;

   symbol1 v=star i=none h=1 w=2 c=black;          /* Actual Observations */
   symbol2 v=none i=spline line=1 w=2 c=red;                /* Extremely high*/
   symbol3 v=none i=spline line=1 w=2 c=blue;               /* Extremely low*/
   symbol4 v=none i=spline line=1 w=2 c=pink;               /* Moderate high */
   symbol5 v=none i=spline line=1 w=2 c=cyan;               /* Moderate low: */
   symbol6 v=none i=spline line=1 w=2 c=lime;               /* Normal*/
   symbol7 v=none i=join line=21 w=1 c=black;

ods graphics on;
ods html;
```

Key SAS code: constructing a funnel plot

```
proc gplot data=temp.usmap;
      title1 j=c h=3 c=black 'Funnel plot. 90-day delinquency rate in credit card loans in the USA';
      axis1 label=(c=black a=90 h=2.5 '90-day delinquency rate') value=(c=black)order=(-0.005 to 0.05 by 0.005) /**/;
      axis2 label=(c=black h=2.5 'Loan volume')  value=(c=black);
            plot dlqrt*volume=1
                  dlqrt_95pct*volume=2
                  dlqrt_5pct*volume=3
                  dlqrt_90pct*volume=4
                  dlqrt_10pct*volume=5
                  dlqrt_us*volume=7
       /overlay  legend=legend1
       vaxis=axis1 haxis=axis2
          cframe = white
       hminor = 0  vminor =0
                   anno=temp.outliers;
      legend1 value=(justify=left) label=('Legend:' justify=left position=(middle left))
                                    value=(h=2.5) frame;
  run;
  ods graphics off;
```

Key SAS code: developing US map

```
/* Developing a risk map */
ods graphics on;
ods html;

proc gmap map=maps.us data=temp.usmap ;
     title1 j=c h=3 c=black 'Chart. Classification of credit card portfolios by 90-day delinquency rate in the USA';

     pattern1 value=solid color=red;          /* Extremely high*/
     pattern2 value=solid color=blue ;        /* Extremely low*/
     pattern3 value=solid color=pink;         /* Moderate high */
     pattern4 value=solid color=cyan;         /* Moderate low*/
     pattern5 value=solid color=lime;    /* Normal: green */
     legend1 value=(justify=left) label=('Risk' justify=left 'levels:' position=(middle left)) frame;
     id  STATECODE ;
     choro risk_level /levels=5 coutline=gray annotate=center legend=legend1;
run;

ods graphics off;
```

Conclusions

Method Limitations:

1. The approach is safe to be applied to homogeneous consumer credit portfolios;
2. If there is an interest in lower level analysis (county), a researcher has to export the results from SAS to the other mapping software (e.g., JMP, ArcGIS).

Refernces

1. David R. Anderson, Dennis J. Sweeney, Thomas A. Williams. Statistics for Business and Economics. 9th edition. Thomson South-Western, 2005
2. Doug Dover. Funnel Plots: Visualizing small area estimates. Presentation at SAS users group meeting in Edmonton, Canada: http://www.sas.com/offices/NA/canada/downloads/presentations/Edmonton2010/Funnel.pdf
3. PROC GMAP, Example 6: Labeling the States on a U.S. Map, http://support.sas.com/onlinedoc/913/docMainpage.jsp